

# Excel Formeln und Funktionen

400 Beispiele, mit Praxislösungen,  
selbst programmieren


IGNATZ SCHELS



Markt+Technik

# KAPITEL 3

## Textfunktionen



Wozu braucht eine Tabellenkalkulation Funktionen, die mit Texten rechnen? Die Frage ist berechtigt, aber wer schon mal Textdaten importiert, mit Artikelnummern gearbeitet oder Verknüpfungen auf Internet-/Intranetseiten gesetzt hat, wird die Vorteile der Textfunktionen zu schätzen wissen.

Hier alle Textfunktionen im Überblick:

<b>Funktion</b>	<b>Erklärung</b>
=CODE()	gibt die Codezahl des Textes aus.
=DM()	wandelt eine Zahl in Text im Währungsformat um.
=FEST()	formatiert eine Zahl als Text mit festen Nachkommastellen.
=GLÄTTEN()	entfernt die Leerzeichen aus einer Zelle.
=GROSS() =GROSS2() =KLEIN()	wandelt Text in Großschrift/Kleinschrift um.
=IDENTISCH()	prüft zwei Zeichenfolgen ab, ob diese identisch sind.
=LÄNGE()	gibt die Länge eines Textes (Anzahl der Buchstaben) aus.
=LINKS()	gibt den linken Teil einer Zelle aus.
=RECHTS()	gibt den rechten Teil der Zelle aus.
=SÄUBERN()	entfernt alle nicht druckbaren Zeichen aus dem Text.
=SUCHEN() =FINDEN()	sucht eine Textstelle im angegebenen Text.
=T()	wandelt die Argumente einer Formel in Text um.
=TEIL()	gibt einen Teil der Zelle aus.
=TEXT()	wandelt eine Zahl in Text um (mit Zahlenformat).
=VERKETTEN()	verknüpft mehrere Texte zu einem Text.
=WECHSELN() =ERSETZEN()	ersetzt einen Text in der Zelle durch einen anderen.
=WERT()	wandelt einen Text in eine Zahl um.
=WIEDERHOLEN()	wiederholt einen Text so oft, wie im zweiten Argument angegeben.
=ZEICHEN()	gibt das Zeichen der angegebenen Codezahl zurück.

**Tabelle 3.1:** Liste der Funktionen in der Kategorie Text

## 3.1 Zahl, Text und Textformat

Unterscheiden Sie im Umgang mit Text zwischen „echten“ Texten und formatierten Texten. Eine Zahl steht immer rechtsbündig in einer Zelle, Texte stehen linksbündig (sofern nicht anders formatiert). Enthält eine Zelle andere Zeichen als die Ziffern 0 bis 9, wird der Inhalt automatisch als Text interpretiert.

Das Zahlenformat der Zelle ändert nichts an der Tatsache, dass die Zelle Zahl oder Text enthält, es liefert nur eine andere Darstellung. Um die Zahl als Text oder mit zusätzlichem Text auszuweisen, formatieren Sie die Zelle mit einem benutzerdefinierten Zahlenformat, das den Text links oder rechts vom Platzhalter für die Zahl anbietet.

Ein Beispiel: Ihre Artikelliste weist die Mengen der Artikel in unterschiedlichen Gebinden aus. Erstellen Sie Zahlenformate für Stück, kg, Liter und Meter:

1. Wählen Sie **FORMAT/ZELLEN/ZAHLEN**.
2. Schalten Sie auf die Kategorie *Benutzerdefiniert*, und tragen Sie diese Zahlenformate ein:

0" Liter"

0" Stück"

0" kg"

0" Meter"

The screenshot shows a Microsoft Excel spreadsheet with the following data:

	A	B	C	D	E
1	Artikelnummer	Bezeichnung	Menge		
2	10-200	Wandfarbe Alpinweiss	10 Liter		
3	10-210	Malerpinsel mittel	5 Stück		
4	10-220	Cips weiss natur	50 kg		
5	10-230	Rauhfaser Tapete weiss	120 Meter		
6					
7					
8					
9					
10					
11					
12					
13					
14					
15					

**Abbildung 3.1:** Zahlen als Text formatiert mit dem richtigen Zahlenformat

### 3.1.1 Das Textformat

Das Textformat verwenden Sie, um eine Zahl zu schützen. Excel hat die meist nützliche, aber manchmal lästige Eigenschaft, eine eingegebene oder importierte Zahl zu interpretieren und diese je nach Beschaffenheit in ein Datum, einen Bruch oder eine Exponentialzahl umzuwandeln:

- Wenn eine Zahl einen Schrägstrich (/) oder einen Bindestrich (-) enthält, wird sie möglicherweise in ein Datumsformat konvertiert, sofern ein gültiges Datum auszumachen ist (2/3 ist der 2. März, 10/5 ist der 10. Mai).
- Wenn eine Zahl einen Doppelpunkt (:) enthält oder auf die Zahl ein Leerzeichen mit dem Buchstaben A oder P folgt, wird sie möglicherweise in ein Uhrzeitformat konvertiert (A und P sind die Kennzeichen für das US-Zeitformat).
- Wenn eine Zahl den Buchstaben E (in Groß- oder Kleinschreibung, zum Beispiel 10e5) enthält, wird sie in Exponentialschreibweise konvertiert.
- Wenn eine Zahl führende Nullen enthält, werden die führenden Nullen weggelassen.

Formatieren Sie die Zellen, die für Zahlen oder Texte dieser Art vorgesehen sind, mit dem Textformat. Enthält die Zelle anschließend eine Zahl, wird diese linksbündig gestellt, bleibt aber trotzdem für Berechnungen aller Art offen:

1. Geben Sie eine Zahl in Zelle A1 ein:

A1: 32


2. Wählen Sie für die Zelle A1 **FORMAT/ZELLEN/ZAHLEN**.

3. Schalten Sie auf die Kategorie *Text* um, und bestätigen Sie mit **OK**.

4. Berechnen Sie die Wurzel aus der Zahl:

B1: =WURZEL(A1)

Die Wurzel wird berechnet, auch wenn die Zahl als Text formatiert ist.




**Abbildung 3.2:** Auch Zahlen im Textformat werden berechnet.

### 3.1.2 Zahl als Text

Wenn Sie ein kleines grünes Dreieck in der linken oberen Ecke der Zelle mit der Textzahl sehen, weist ein Smarttag darauf hin, dass die Zelle eine Zahl enthält, die Excel nicht als solche interpretieren kann. Das kommt hauptsächlich bei importierten Daten vor oder bei Verknüpfungen auf externe Daten oder Internet-/Intranetseiten.

Sie können diese Zellwerte zwischen Zahl und Text ganz einfach in echte Zahlen umwandeln. So gehen Sie vor:

1. Markieren Sie den Zellbereich, der die Textzahlen enthält. Wenn die Zahlen auf mehrere Bereiche verteilt sind, wählen Sie **BEARBEITEN/GEHEZU/INHALTE/KONSTANTEN**.
2. Klicken Sie auf den Pfeil am Smarttag, und wählen Sie **IN EINE ZAHL UMWANDELN**.




**Abbildung 3.3:** Die Textzahlen werden in echte Zahlen umgewandelt.

In früheren Versionen, die noch keine Smarttags kannten, lassen sich die Textzahlen mit einem einfachen Trick in Zahlen verwandeln:

1. Schreiben Sie in eine freie Zelle die Zahl 1.
2. Markieren Sie die Zelle, und kopieren Sie den Inhalt mit **[Strg] + [C]** in die Zwischenablage.
3. Markieren Sie alle Textzahlen mit **BEARBEITEN/GEHEZU/INHALTE/KONSTANTEN**.
4. Wählen Sie **BEARBEITEN/INHALTE EINFÜGEN/MULTIPLIZIEREN**.

Die markierten Zellen werden mit der Zahl 1 aus der Zwischenablage multipliziert und damit in echte Zahlen verwandelt.



**Abbildung 3.4:** Zahlenumwandlung: Texte einfach mit 1 multiplizieren

## 3.2 Rechnen mit Text

Textfunktionen rechnen nicht im mathematischen, arithmetischen Sinn mit Texten in der Tabelle, sondern analysieren diesen meist buchstabenweise, als Zeichenfolge. In der Praxis lassen sich damit komplexe Aufgaben meistern:

### Rechnen mit Buchstaben

Zahlen, die in Textform oder als Teil eines Textes vorliegen, können für Berechnungen verwendet werden, wenn sie aus der Textkette „herausgerechnet“ werden. Funktionen wandeln Textzahlen in Zahlen um und umgekehrt auch Zahlen in Buchstaben.

### Textlayout

Texte, die aus anderen Quellen stammen, werden fachgerecht für das Tabellenlayout aufbereitet, überflüssiger Text wird entfernt, Texte werden neu zusammengestellt oder miteinander verknüpft. Groß-/Kleinschrift lässt sich wechseln.

### Textanalyse

Texte in Zellen werden auf ihren ANSI-Wert (Zahlencode) überprüft, Texte werden verglichen und ausgetauscht, Formeln suchen nach Textteilen und wechseln diese automatisch aus.

### 3.3 Die Funktionen LINKS() und RECHTS()

Mit diesen Funktionen berechnen Sie den linken bzw. rechten Teil eines Textes.

=LINKS(Text;Anzahl\_Zeichen)

=RECHTS(Text;Anzahl\_Zeichen)

Text ist der Text oder die Zelle mit dem Text. Im Argument *Anzahl\_Zeichen* geben Sie eine Zahl ein. Diese gibt an, wie viele Zeichen von links bzw. rechts im Ergebnis stehen sollen:

A1: ABCDEFG

A2: =LINKS(A1;1) Ergebnis: A

A3: =LINKS(A1;2) Ergebnis: AB

A4: =RECHTS(A1;3) Ergebnis: EFG

Hier eine Artikelnummertabelle, die links den dreistelligen Anbietercode enthält und rechts mit fünf Ziffern den Preis des Artikels. Die Textfunktionen LINKS() und RECHTS() schneiden diese Textstellen heraus:

A1: Artikelnummer

B1: Anbieter

C1: Preis

A2: ABC-0010-2299

=LINKS(A2;3)

=RECHTS(A2;5)

Ergebnis:

ABC

02299

#### 3.3.1 Praxisbeispiel: EAN-Nummern analysieren

Die internationale EAN-Organisation stellt ein Nummernsystem für EDV-Systeme und Strichcodes zur Verfügung. Das EAN-System (european article number) kennzeichnet Waren eindeutig und macht sie kompatibel für Lese- und Verwaltungssysteme aller Art.

Sie erhalten von Ihrem Zulieferer eine Artikelliste, in der die Warengruppen und die EAN-Nummer zusammengefasst sind. Erstellen Sie je eine Spalte für die Warengruppe und die EAN-Nummer:

■ Warengruppe: erste Ziffer der Artikelnummer

■ EAN-Nummer: Ziffern

1. Markieren Sie die Spalte B, und fügen Sie mit Strg + + zwei neue Spalten ein.
2. Berechnen Sie in der neuen Spalte B die Warengruppe aus dem ersten Buchstaben der Artikelnummer:

B2: =LINKS(A2;1)

3. Berechnen Sie in Spalte C die EAN-Nummer aus den letzten Buchstaben der Artikelnummer:

C2: =RECHTS(A2;13)






Abbildung 3.5: Die Artikelliste mit Artikelnummer




Abbildung 3.6: Die beiden Nummern sind mit Hilfe von Textfunktionen ermittelt.

### 3.4 Die Funktion TEIL()

Die Funktionen LINKS() und RECHTS() analysieren, wie im Beispiel gezeigt, einen Text von links oder rechts. In Fällen, in denen die gesuchte Textmenge mitten in einem Text beginnt, greifen Sie auf die Funktion TEIL() zurück:

=TEIL(Text;Erstes\_Zeichen;Anzahl\_Zeichen)

Text ist der Text, der analysiert wird. Mit dem Argument *Erstes\_Zeichen* geben Sie die Position an, an der die neue Textkette beginnt, und *Anzahl\_Zeichen* gibt an, wie lang diese Textkette sein soll (Anzahl Buchstaben).

A1: ABCDEFG

B1: =TEIL(A1;2;3) Ergebnis: BCD

### 3.4.1 Praxisbeispiel: Teilenummern analysieren

Größere Betriebe haben in der Praxis ein Nummernsystem für alle Gegenstände, die zum Inventar der Firma gezählt oder die in der Firma gefertigt werden. Diese Nummern beinhalten neben einer Zählnummer auch Änderungsnummern und Prüfciffern. Hier ein Beispiel: Das Nummernsystem unterscheidet zwischen Kennziffer, Konstruktionsgruppe, Identifikationsnummer und Änderungskennzeichen:




Abbildung 3.7: Ein Nummernsystem mit 5 Teilenummern

Ihre Aufgabe besteht darin, eine Artikelnummernliste zu analysieren, die einzelnen Teile herauszulesen und in Spalten zu übertragen. Anschließend können Sie die Nummern nach Kriterien wie Konstruktionsgruppe oder Änderungsnummer sortieren und filtern. Diese Artikelliste liegt vor:

	A	B	C	D
1	Teilenummer	Bezeichnung	Empf. VKPreis	Liefermenge
2	W 011 799 11 111	Kühlergrill	199,00	12
3	A 023 810 03 111	Ölfilter	12,99	23
4	A 023 810 04 111	Zylinderkopfdichtung	89,99	120
5	A 300 900 10 111	Keilriemen	8,90	120
6	A 123 002 20 123	Wasserpumpe	211,00	3
7	W 012 789 12 123	Scheibenwischermotor	189,00	55
8	W 011 800 11 123	Vergaser	120,00	56

**Abbildung 3.8:** Analysieren Sie diese Teilenummern nach den einzelnen Gruppen.

Kennziffer und Änderungskennzeichen können Sie mit den Funktionen LINKS() und RECHTS() herausrechnen, für die restlichen Gruppen brauchen Sie die Funktion TEIL(). Achten Sie auf die Leerzeichen zwischen den Gruppen, diese müssen natürlich berücksichtigt werden:

1. Fügen Sie ab Spalte B fünf neue Spalten ein.
2. Schreiben Sie die Überschriften und die Formeln zur Berechnung der einzelnen Gruppen:

B2: =LINKS(A2;1) oder =TEIL(A2;1;1)

C2: =TEIL(A2;3;3)

D2: =TEIL(A2;7;3)

E2: =TEIL(A2;11;2)

F2: =TEIL(A2;14;3)

3. Kopieren Sie die Formeln mit dem Füllkästchen nach unten auf die übrigen Zeilen der Artikelliste.

The screenshot shows an Excel spreadsheet with the following data:

	A	B	C	D	E	F	G
1	Nummer	Kennziffer	Teile-Nummer	Konstruktionsgruppe	Identifikations-Nummer	Änderungs-Kennzeichen	Bezeichnung
2	W 011 799 11 111	W	011	799	11	111	Kühlergrill
3	A 023 810 03 111	A	023	810	03	111	Ölfilter
4	A 023 810 04 111	A	023	810	04	111	Zylinderkopfdi
5	A 300 900 10 111	A	300	900	10	111	Keilriemen
6	A 123 002 20 123	A	123	002	20	123	Wasserpumpe
7	W 012 789 12 123	W	012	789	12	123	Scheibenwische
8	W 011 800 11 123	W	011	800	11	123	Vergaser
9							
10							
11							
12							
13							
14							
15							

Abbildung 3.9: Die Artikelnummer ist in Einzelteile zerlegt.

### 3.5 Die Funktionen FINDEN() und SUCHEN()

Diese beiden Funktionen werden meist in Kombination mit anderen Textfunktionen verwendet. Sie liefern die Information darüber, wo ein bestimmtes Zeichen oder eine Teiltextkette zu finden ist, und diese Information schachteln Sie in der Praxis in eine weitere Funktion, mit der Sie die Teiltextkette ausgeben.

=FINDEN(Suchtext;Text;Erstes\_Zeichen)

=SUCHEN(Suchtext;Text;Erstes\_Zeichen)

Das Argument *Text* steht für den Text, in dem Sie etwas suchen, und *Suchtext* ist der zu suchende Text. Beide Argumente können Sie aus Zellen oder Bereichsnamen beziehen oder direkt eingeben, dazu verwenden Sie aber Anführungszeichen. Mit dem Argument *Erstes\_Zeichen*, das nicht unbedingt angegeben werden muss, bestimmen Sie, an welchem Zeichen des Textes die Suche beginnen soll. Lassen Sie das Argument weg, beginnt die Suche am ersten Zeichen.

Das Ergebnis der Funktion ist eine Zahl, nämlich die Position des Suchtextes.

A1: ABCDEFG  
A2: =FINDEN("A";A1) Ergebnis: 1  
A3: =FINDEN("B";A1) Ergebnis: 2  
A4: =FINDEN("CD";A1) Ergebnis: 3

Die Funktion SUCHEN() ist identisch mit der Funktion FINDEN() mit einer Ausnahme: FINDEN() berücksichtigt die Groß- und Kleinschreibung des Suchbegriffs.

## 3.6 Die Funktion LÄNGE()

Mit dieser Funktion ermitteln Sie die Länge eines Textes bzw. eines Zelleintrags. LÄNGE() kann auch die Länge eines Zahlenwerts ausgeben, in erster Linie wird die Funktion aber für Berechnungen von und in Textketten verwendet.

=LÄNGE(Text)

Das erste und einzige Argument *Text* erhält die Zelle, in der sich der zu messende Text befindet, oder einen Text in Anführungszeichen.

A1: ABCDEFG  
A2: =LÄNGE(A1) Ergebnis: 7  
A3: =LÄNGE("Hallo Deutschland!") Ergebnis: 18

### 3.6.1 Praxisbeispiel: Vornamen und Nachnamen trennen

Ihre Personalliste liefert die Namen der Mitarbeiter in einer Zelle, Vorname und Nachname sind nicht getrennt. Erleichtern Sie Ihrem Personalsachbearbeiter die Arbeit und trennen Sie die Namen, damit er die Liste bei Bedarf auch nach Nachnamen sortieren kann.

1. Schreiben Sie in B2 die Funktion, die das Leerzeichen zwischen Vor- und Nachnamen findet:

B2: =FINDEN(" ";B2)

2. Schachteln Sie die Funktion in eine LINKS()-Funktion ein, die das Ergebnis als Längenangabe für die Teilkette benutzt:

B2: =LINKS(A2;FINDEN(" ";A2)-1)

3. Schreiben Sie in Zelle C2 die Funktion, die den Nachnamen ab der Position des Leerzeichens ausgibt:

=TEIL(A2;FINDEN(" ";A2)+1;LÄNGE(A2)-FINDEN(" ";A2))

Microsoft Excel - Formeln und Funktionen Beispiele.xls

Frage hier eingeben

Arial 10

	A	B	C	D	E	F
1	<b>Name</b>	Vorname	Nachname			
2	Friedrich Ullmann					
3	Doris Grabowski					
4	Monika Wells					
5	Egon Wallner					
6	Birgit Lichtner					
7	Bruno Jörgens					
8						
9						
10						
11						
12						
13						

Bereit

Abbildung 3.10: Vorname und Nachname müssen getrennt werden.

B2 =LINKS(A2;FINDEN(" ";A2)-1)

	A	B
1	<b>Name</b>	Vorname
2	Friedrich Ullmann	Friedrich
3	Doris Grabowski	Doris
4	Monika Wells	Monika
5	Egon Wallner	Egon
6	Birgit Lichtner	Birgit
7	Bruno Jörgens	Bruno

C2 =TEIL(A2;FINDEN(" ";A2)+1;LÄNGE(A2)-FINDEN(" ";A2))


	A	B	C
1	<b>Name</b>	Vorname	Nachname
2	Friedrich Ullmann	Friedrich	Ullmann
3	Doris Grabowski	Doris	Grabowski
4	Monika Wells	Monika	Wells
5	Egon Wallner	Egon	Wallner
6	Birgit Lichtner	Birgit	Lichtner
7	Bruno Jörgens	Bruno	Jörgens

Abbildung 3.11: Vor- und Nachname sind getrennt.

### 3.6.2 Praxisbeispiel: Minuszeichen rechts von der Zahl

In SAP-Berichten taucht häufig dieser Layoutfehler auf, auch andere Abrechnungssysteme auf Host-Rechnern liefern ihn: Das Minuszeichen steht nicht wie üblich links an der Zahl, sondern rechts davon. Lässt sich der Export nicht im Quellsystem reparieren, greifen Sie zur Excel-Formel und ändern Ihre Tabellen.

Bei importierten Tabellen oder Textdaten wird der Fehler optisch sofort erkennbar, die falschen Minuszahlen werden linksbündig angeordnet, weil Excel sie als Text interpretiert.



**Abbildung 3.12:** Das Minuszeichen steht fälschlicherweise rechts an der Zahl.

1. Setzen Sie den Zellzeiger rechts neben die erste Zahl in der Spalte C.
2. Schreiben Sie diese Formel:  

$$=WENN(RECHTS(C2;1)="-";LINKS(C2;LÄNGE(C2)-1)*-1;C2)$$
3. Kopieren Sie die Formel per Doppelklick auf das Füllkästchen nach unten bis zur letzten Zeile.


Die Formel prüft über eine WENN-Funktion, ob rechts vom Zellinhalt ein Minuszeichen steht, und schneidet in diesem Fall den Inhalt um eine Ziffer ab. Um die Zahl als negativ auszuweisen, wird der Formelteil mit  $-1$  multipliziert:

$\text{LINKS}(C2;\text{LÄNGE}(C2)-1)*-1$

Hat die Zelle kein Minuszeichen am rechten Rand, gibt die WENN-Funktion den Zellinhalt ohne Änderung aus. Damit erhalten Sie eine korrekte Liste mit positiven und negativen Zahlen.

### Tip

Ersetzen Sie die falschen Zahlen gleich durch die berechneten Werte: Markieren Sie die Formeln, und ziehen Sie den Rand der Markierung mit gedrückter rechter Maustaste auf die Spalte C. Wählen Sie aus dem Kontextmenü *Hierhin nur als Werte kopieren*.



**Abbildung 3.13:** Die Formeln werden gleich als Werte auf die alten Zahlen kopiert.



### 3.6.3 Praxisbeispiel: Quersumme berechnen

Die Funktion TEIL() lässt sich auch zuverlässig zur Ermittlung der Quersumme einer Zahl verwenden. Wenn die Größe der Zahl bekannt ist, können Sie die einzelnen Ziffern addieren:

```
A1: 1234
A2: =TEIL(A1;1;1)+TEIL(A1;2;1)+TEIL(A1;3;1)+TEIL(A1;4;1)
Ergebnis: 10
```

Schwieriger wird es, wenn die Größe der Zahl nicht bekannt ist. Hier wäre eine benutzerdefinierte Funktion hilfreich. Wer sich noch nicht an die VBA-Programmierung wagen will, kann die Aufgabe aber auch per Formel lösen:

Die Funktion LÄNGE() prüft zunächst die Länge der Zahl ab:

```
A1: 12345
A2: =LÄNGE(A1)
```

Mit der Funktion WAHL() geben Sie für jede Länge je ein Argument an. Das zweite Argument bildet das Ergebnis, wenn die Länge 1 ist, das dritte, wenn Länge=2 usw...

```
=WAHL(LÄNGE(A1);arg1;arg2;arg3; ... argn)
```

Der Rest ist Schreiarbeit, für jedes Argument berechnen Sie die Quersumme über die TEIL-Funktionen (hier für maximal 5-stellige Zahlen):

```
=WAHL(LÄNGE(A1);
A1;
TEIL(A1;1;1)+TEIL(A1;2;1);
TEIL(A1;1;1)+TEIL(A1;2;1)+TEIL(A1;3;1);
TEIL(A1;1;1)+TEIL(A1;2;1)+TEIL(A1;3;1)+TEIL(A1;4;1);
TEIL(A1;1;1)+TEIL(A1;2;1)+TEIL(A1;3;1)+TEIL(A1;4;1)+TEIL(A1;5;1))
```

Die VBA-Funktion sollten Sie natürlich auch kennen lernen, sie berechnet die Quersumme für beliebig große Zahlenwerte:

```
Function QuerSumme(strText As String) As Integer
Application.Volatile
Dim i As Integer
For i = 1 To Len(strText) QuerSumme = QuerSumme + CInt(Mid(strText, i, 1))
Next i
End Function
```

**Listing 3.1:** Benutzerdefinierte Funktion für Quersummen zu beliebig großen Zahlen

## 3.7 Die Funktion VERKETTEN()

Das ist eigentlich eine überflüssige Funktion, sie wird selten benutzt, weil es für die Verkettung von Texten eine einfachere Methode gibt:

**=VERKETTEN(Text1;Text2; ... Textn)**

In den Funktionsklammern können Sie mit Semikolon als Trennzeichen beliebig viele zu verkettende Texte angeben. Das können direkte Zeichen in Anführungszeichen sein, Zellbezüge oder Bereichsnamen. Das Ergebnis der Funktion ist eine Textkette aus den einzelnen Argumenten:

A1: "Heute ist "

A2: "Montag"

A3: =VERKETTEN(A1;A2) Ergebnis: "Heute ist Montag"

In der Praxis wird an Stelle der Funktion das &-Zeichen verwendet, es erfüllt denselben Zweck und macht die Formeln etwas übersichtlicher:

=A1&A2

Wenn Sie ein Leerzeichen in den Text einbinden wollen, schreiben Sie dieses in Anführungszeichen:

=A1&" "&A2

Besonders nützlich ist die Textverkettung, wenn Sie Zelleninhalte mit Zeilenumbruch versehen wollen. Hier ein Beispiel:

Ihre Tabelle enthält mehrere Sätze, die Sie zu einem Text zusammenführen wollen. Mit jedem Satz sollte aber eine neue Zeile beginnen.

	A
1	Bitte legen Sie die CD ein.
2	Starten Sie die Installation mit SETUP.EXE.
3	Befolgen Sie die Anweisungen des Assistenten.

**Abbildung 3.14:** Diese drei Sätze sollen zu einem Text verkettet werden.

1. Schreiben Sie die erste Kette:

=A1&" "

2. Drücken Sie `[Alt] + [↵]`, um einen Zeilenumbruch in die Formel einzubauen.
3. Schreiben Sie die zweite Kette, drücken Sie nach dem "-"Zeichen wieder `[Alt] + [↵]`, und hängen Sie den dritten Satz an.

```
=A1&"
"&A2&"
"&A3
```

Alternativ zu dieser Technik können Sie auch das ANSI-Zeichen für den Zeichenumbruch in die Formel einbauen. Es lautet ZEICHEN(10):

```
=A1&ZEICHEN(10)&A2&ZEICHEN(10)&A3
```



Abbildung 3.15: Zwei Möglichkeiten, Zeilenumbrüche mit zu verketten

### 3.7.1 Praxisbeispiel: Organigramm beschriften

Sie haben die Aufgabe, das Organigramm Ihres Projektteams zu erstellen, wenn möglich natürlich in Excel. Die Zeichnung sollte aber so flexibel sein, dass eine Änderung in der Tabelle automatisch in die Zeichnung eingetragen wird. Erstellen Sie zunächst die Liste der Projektmitarbeiter mit Nummer, Name und Position.


	A	B	C	D
1	<b>Nr</b>	<b>Vorname</b>	<b>Name</b>	<b>Position</b>
2	1	Dr. Friedrich	Bergmeister	Projektleiter
3	1.1	Werner	Schumann	Planung & Bau
4	1.2	Monika	Dietrich	Finanzen
5	1.3	Paul	Schröder	Organisation

Abbildung 3.16: Das Projektteam ist aufgestellt.

**Tipp**

Die Gliederungsnummern schreiben Sie am besten mit einem vorangestellten Apostroph oder in eine Spalte mit Textformatierung, sonst produziert Excel wieder ungewollte Datumswerte.

Zeichnen Sie anschließend ein Organigramm. Die Symbolleiste *Zeichnen* hält unter der Kategorie *AutoFormen* schöne Werkzeuge wie Verbindungslinien und geometrische Objekte bereit.



**Abbildung 3.17:** Organigramm zeichnen und mit Verbindungslinien versehen

Um die Kästchen variabel zu beschriften, müssen Sie die Informationen aus der Liste zunächst zusammenfassen und dann mit den grafischen Objekten verknüpfen. Bei dieser Gelegenheit arbeiten Sie Zeilenumbrüche ein:

- Schreiben Sie in die erste Namenszeile der Spalte E eine Formel, die zuerst die Nummer und die Position, anschließend den Vornamen und den Namen des Mitarbeiters in eine Textkette verpackt. Damit Vorname und Name in einer neuen Zeile stehen, klemmen Sie die Funktion ZEICHEN(10) dazwischen.


**Hinweis**

E2: =A2&" "&D2&ZEICHEN(10)&B2&" "&C2

Die Codenzahl 10 steht für das Zeichen „Zeilenumbruch“. Dieses Zeichen ist nicht sichtbar, wird aber gedruckt und funktioniert auch in grafischen Objekten.

- Kopieren Sie die Formel auf die übrigen Mitarbeiterzeilen. Markieren Sie das erste Kästchen, und klicken Sie in die Bearbeitungsleiste. Schreiben Sie die Verknüpfung zur Formel in die Leiste, klicken Sie auf den Bezug, um die Adresse absolut zu machen: =E\$2
- Verfahren Sie so auch mit den anderen gezeichneten Kästchen, verknüpfen Sie diese mit den Formeln, die alle Informationen zu einer Textkette zusammenfassen.

Damit ist das variable Organigramm fertig, Sie können die Spalte E ausblenden und in die Spalten A bis D beliebig neue Werte eintragen, die durch die Verknüpfung automatisch wieder korrekt in die Organigramm-Objekte eingesetzt werden.



**Abbildung 3.18:** Das variable Organigramm mit Zeilenumbrüchen

### 3.8 Die Funktion IDENTISCH()

Der Name der Funktion drückt aus, für welche Aufgaben sie sich eignet: Mit IDENTISCH() prüfen Sie ab, ob zwei Zelleinträge genau gleich sind. Das ist besonders in größeren Listen von Vorteil, da die Formel schnell die Voraussetzung schaffen kann für Filter- und Sortieraufgaben. Hier ein Beispiel:

1. Schreiben Sie die Funktion zur Erzeugung einer Zufallszahl zwischen 100 und 150 in die Zelle A1. Verwenden Sie die Analyse-Funktion ZUFALLSBEREICH(), müssen Sie das Add-In installiert haben (siehe *Kapitel 12*).

A1: =ZUFALLSBEREICH(100;150)

2. Kopieren Sie die Formel nach unten bis zur Zelle A20 und eine Spalte weiter in den Zellbereich B1:B20.
3. Schreiben Sie die Funktion, die überprüft, ob die beiden Zahlen in einer Zeile identisch sind:

C1: =IDENTISCH(A1;B1)

4. Kopieren Sie die Formel nach unten bis zu Zeile 20. Mit jeder Neuberechnung der Tabelle mit  erhalten Sie jetzt die Wahrheitswerte WAHR (beide Zahlen identisch) oder FALSCH (Zahlen nicht identisch). Ergänzen Sie die Formel noch mit einer WENN-Funktion, die nur bei identischen Werten einen \* ausgibt:




Abbildung 3.19: Sind die beiden Zufallszahlen identisch?

## 3.9 Die Funktion WECHSELN()

Mit dieser Funktion ersetzen Sie einen Text durch einen neuen Text. Das ist vor allem bei importierten Texten wichtig und nützlich, wenn diese nicht in gewünschter Form vorliegen. Auch für die Konvertierung fremdsprachiger Texte ist WECHSELN() unerlässlich.

=WECHSELN(Text;Alter\_Text;Neuer\_Text;Ntes\_Auftreten)

Das Argument *Text* steht für eine Zelle oder einen Text in Anführungszeichen. Mit *Alter\_Text* wird die Textstelle bezeichnet, die zu wechseln ist, und *Neuer\_Text* enthält den Text, der an dessen Stelle tritt. Mit *Ntes\_Auftreten* können Sie angeben, welcher von mehreren gefundenen Texten ersetzt werden soll. Wenn Sie das Argument nicht besetzen, wird jede Textstelle ersetzt, geben Sie beispielsweise eine 2 an, wird nur der 2. auftretende Text (von links gerechnet) ersetzt.

A1: ABCDEFG

A2: =WECHSELN(A1;"ABC";"123") Ergebnis: 123DEFG

### 3.9.1 Praxisbeispiel: Fremdwährungen – Punkte durch Kommas ersetzen

Textdateien mit Beträgen in Fremdwährungen sind häufig problematisch beim Import: Wer nicht aufpasst, erhält einige der mit Punkt als Dezimaltrennzeichen notierten Beträge als Datumswerte.

Die Preisliste für Irish Malt Whisky vom Whisky Store ist angekommen, leider aber im Textformat, und die Beträge sind mit Punkt statt Komma als Dezimalstelle ausgewiesen:




Abbildung 3.20: Preisliste im Textformat

- Öffnen Sie die Textdatei mit DATEI/ÖFFNEN, schalten Sie im Feld *Dateityp* auf *Text* um.
- Der Text-Assistent wird aktiv, schalten Sie auf *Getrennt* um, und geben Sie das Semikolon als Trennzeichen an.
- Bestätigen Sie alle weiteren Abfragen, werden die Preise mit Punkt als Trennzeichen eingelesen. Mit der ERSETZEN-Funktion wandeln Sie die Punkte in Kommas um, und die Preise sind korrekt:

G2: =WECHSELN(F2;". ";",")

- Die Zahlen können jetzt zwar als solche benutzt werden, Excel behandelt sie aber noch als Text, was an der linksbündigen Formatierung zu erkennen ist. Mit der Funktion WERT() wandeln Sie die Texte in Zahlen um:

=WERT(WECHSELN(F2;". ";","))

The screenshot shows an Excel spreadsheet with the following data:

1	Marke	Beschreibung	Alter	Vol%	Inhalt	Preis	Preis korrekt
2	Bushmills	Three Woods	16	40%	0.7	56.90	56,90
3	Bushmills	Malt, Madeira Finish, einzelnummeriert	21	40%	0.7	148.50	148,50
4	Bushmills	Single Rum Cask	14	54.40%	0.7	118.50	118,50
5	Bushmills	Single Sherry Cask	14	53.70%	0.7	107.50	107,50
6	Connemara	Peated Single Malt, GP	12	41%	0.7	59.90	59,90
7	Connemara Cae	Peated Single Malt, schlanke Flasche	12	ca. 60%	0.7	32.90	32,90
8	Connemara	Peated Single Malt, GP	12	40%	0.7	24.90	24,90
9	Connemara Cae	Peated Single Malt, bauchige Flasche	12	58.90%	0.7	36.90	36,90
10	Locke's	Keramikkrug	8	40%	0.7	47.90	47,90
11	Magilligan	Peated Single Malt	8	43%	0.7	33.90	33,90
12	Tyrconnell	5 Sterne, Geschenkpackung mit 2 Gläsern	8	40%	0.7	27.90	27,90
13	Tyrconnell	in Tube	8	40%	0.7	19.90	19,90

Abbildung 3.21: Mit der Funktion WECHSELN() Punkte in Kommas umtauschen

### Tipp

Der Text-Assistent bietet auch die Möglichkeit, die Dezimalzeichen von Zahlenspalten umzudefinieren. Klicken Sie in der letzten Abfrage auf die Schaltfläche *Werte*, und tragen Sie den Punkt als Dezimalzeichen ein.



### 3.9.2 Praxisbeispiel: Umlaute und Sonderzeichen austauschen

Mit der Funktion WECHSELN() lassen sich Texte schnell „umprogrammieren“. Sie können Sonderzeichen entfernen, Umlaute gegen Doppelvokale tauschen oder fremdsprachige Zeichen aus dem Windows-Zeichensatz eintragen. Um mehr als eine Textstelle auszutauschen, schachteln Sie die WECHSELN()-Funktion mehrfach. Hier ein Beispiel:

Schreiben Sie eine Funktion, die alle „ä“ durch „ae“, alle „ö“ durch „oe“ und „ü“ durch „ue“ ersetzt:

A1: Bürgerbräu Wörth

B1: =WECHSELN(WECHSELN(WECHSELN(A1;"ö";"oe");"ä";"ae");"ü";"ue")

Wenn Sie eine größere Liste mit Ersatzbegriffen versehen wollen, schreiben Sie diese Liste in einen freien Bereich der Tabelle und verwenden die Zelladressen aus der Liste in der Funktion:




Abbildung 3.22: Funktion mehrfach schachteln für mehrere Wechsel

### 3.9.3 Praxisbeispiel: Zeilenumbrüche entfernen

Enthält eine Zelle Zeilenumbrüche, kann das zwei Ursachen haben: Entweder hat die Zelle eine entsprechende Formatierung, oder der Text enthält ein Zeichen, das den nachfolgenden Text in die nächste Zeile zwingt:

- Unter **FORMAT/ZELLEN/AUSRICHTUNG** wird die Option *Zeilenumbruch* gesetzt, damit der Text in der Zelle mehrzeilig geschrieben wird.
- Einen Zeilenumbruch im Text erzeugen Sie mit **[Alt] + [↵]** oder mit der Funktion **=ZEICHEN(10)**.

#### Hinweis

Wenn Sie in einem Text einen Zeilenumbruch schreiben, wird diesem nach Abschluss mit der **[↵]**-Taste automatisch die Formatierung unter **FORMAT/ZELLEN** zugewiesen.

Mit dieser Funktion entfernen Sie alle Zeilenumbrüche aus der Zelle A1 und ersetzen sie durch je ein Leerzeichen:

```
=WECHSELN(A1;ZEICHEN(10);" ")
```

## 3.10 Die Funktion ERSETZEN()

Diese Funktion ist der Funktion **WECHSELN()** ähnlich, bietet aber noch zusätzlichen Komfort. Das zweite Argument bietet die Möglichkeit, die Position zu bestimmen, an der die Ersetzung beginnen soll, und im dritten Argument geben Sie die Anzahl Zeichen an:

```
=ERSETZEN(Alter_Text;Erstes_Zeichen;Anzahl_Zeichen;Neuer_Text)
```

Das Argument *Alter\_Text* bezeichnet die Zelle oder den Text in Anführungszeichen, in dem etwas ersetzt wird.

Mit *Erstes\_Zeichen* geben Sie die Position des Zeichens in *Alter\_Text* an, an der die Ersetzung beginnen soll.

Das Argument *Anzahl\_Zeichen* bestimmt, wie viele Zeichen aus *Alter\_Text* durch *Neuer\_Text* ersetzt werden sollen.

A1: Hans Meier

A2: =ERSETZEN(A1;1;4;"Rudi")      Ergebnis: Rudi Meier

### 3.11 Die Funktionen GROSS(), GROSS2() und KLEIN()

Wandeln Sie mit diesen Funktionen alle Kleinbuchstaben eines Textes in Großbuchstaben um oder umgekehrt alle Großbuchstaben in Kleinbuchstaben.

=GROSS(Text)

=GROSS2(Text)

=KLEIN(Text)

Das Argument *Text* wird mit einem Zellbezug, einem Bereichsnamen oder einem Text in Anführungszeichen besetzt, das Ergebnis der Funktion ist der umgewandelte Text.

A1: berchtesgaden

A2: =GROSS(A1) Ergebnis: BERCHTESGADEN

A3: HAMBURG

A4: =KLEIN(A3) Ergebnis: hamburg

Verwenden Sie die Funktion GROSS2(), um in einem kleingeschriebenen Text alle Wörter großzuschreiben.

A1: mit mwst

A2: =GROSS2(A1) Ergebnis: Mit Mwst

In der Praxis werden häufig Firmennamen aus Textdateien importiert, wobei die korrekte Schreibweise verloren geht. Testen Sie die Konvertierfunktionen für Texte an dieser Liste:

AMERITECH

BELGACOM

Bell Atlantic

Bell Canada

Bell South

BEZEQ

British petrol

DBP Telekom

FRANCE TELECOM

gte limited

HONG KONG Telecom




Abbildung 3.23: Textkonvertierung Groß- und Kleinschrift

## 3.12 Die Funktion GLÄTTEN()

Diese Funktion gehört zu den wichtigsten Textfunktionen für Anwender, die viel mit Textimport und Textdaten aus Großrechnersystemen zu tun haben. Nicht alle Systeme können Texte in eine für Excel lesbare Form exportieren, und die Ergebnisse bereiten den Anwendern häufig viel Arbeit. Das gilt besonders für so genannte Druckdateien, Daten, die für die Ausgabe auf einen Drucker aufbereitet und als Dateien ausgegeben werden. Werden diese in Excel eingelesen, sind sie meist mit zahlreichen Leerzeichen, Sonderzeichen und anderem „Datenmüll“ behaftet. Die Funktion GLÄTTEN() schafft Abhilfe:

=GLÄTTEN(Text)

Das Argument *Text* steht für eine Zelle, einen Bereichsnamen oder einen Text in Anführungszeichen. Das Ergebnis der Funktion ist der Text ohne führende und folgende Leerzeichen, und alle Leerzeichen innerhalb des Textes sind auf jeweils eines reduziert.

A1:   Wareneingang   1. Halbjahr

A2: =GLÄTTEN(A1)

Ergebnis: Wareneingang 1. Halbjahr

### 3.12.1 Praxisbeispiel: Mitarbeiterliste säubern

Die aus dem Personalabrechnungssystem importierte Personalliste ist nicht besonders glücklich formatiert, in jeder Zelle sind überflüssige Leerzellen zu finden.

	A	B	C
1	<b>Name (Mitarbeiter)</b>	<b>Vorname</b>	<b>Standort (Stadt)</b>
2	Meier	Hans- Hubert	München
3	Huber	Rudolf	Stuttgart
4	Müller	Beate	Hamburg
5	Friedrichs	Karl Heinz	nicht bekannt
6	von Klausewitz	Günther	Stuttgart
7	Dr. Elbinger	Franz	Mannheim

**Abbildung 3.24:** Personalliste mit Leerzeilen.

Erstellen Sie eine Matrix, in der Sie die Liste neu abbilden und dabei mit der Funktion `GLÄTTEN()` die überflüssigen Leerzeichen entfernen:

1. Markieren Sie einen gleich großen Bereich für die Zielliste, zum Beispiel A10:C16.
2. Schreiben Sie diese Funktion in die aktive Zelle:  
`=GLÄTTEN(A1:C7)`
3. Drücken Sie zum Abschluss `[Strg] + [↕] + [↩]`, um die Formel als Matrixformel auszuweisen.

Die Matrix können Sie nur geschlossen bearbeiten, und nach jeder Änderung müssen Sie wieder mit `[Strg] + [↕] + [↩]` abschließen.

#### Hinweis

Alternativ zu `GLÄTTEN()` können Sie auch die Funktion `WECHSELN()` verwenden, um alle Leerzeichen zu entfernen, auch die einzelnen innerhalb des Textes.

Die größten Probleme bereiten Zellen, die nur Leerzeichen enthalten, und hier verweigert auch die Funktion `GLÄTTEN()` ihre Dienste. Wenn der Zellinhalt nur aus Leerzeilen besteht, bleibt das Funktionsergebnis leer.

`=GLÄTTEN(A1)`

ergibt keinen Inhalt, wenn in A1 z. B. 10 Leerzeichen stehen. Um zu prüfen, ob und wie viele Zeichen sich in einer Zelle befinden, verwenden Sie die Funktion `LÄNGE()`. Wenn Sie diese in einer Hilfsspalte anwenden, finden Sie heraus, ob die Zelle leer ist (Länge 0) oder eine Reihe von Leerzeichen enthält.

The screenshot shows a Microsoft Excel window titled "Microsoft Excel - Formeln und Funktionen Beispiele.xls". The formula bar displays the formula `=GLÄTTEN(A1:C7)`. The active cell is A10, which contains the text "Name (Mitarbeiter)". The table below shows the following data:

	A	B	C	D	E	F
1	Name (Mitarbeiter)	Vorname	Standort (Stadt)			
2	Meier	Hans- Hubert	München			
3	Huber	Rudolf	Stuttgart			
4	Müller	Beate	Hamburg			
5	Friedrichs	Karl Heinz	nicht bekannt			
6	von Klausewitz	Günther	Stuttgart			
7	Dr. Elbinger	Franz	Mannheim			
8						
9						
10	Name (Mitarbeiter)	Vorname	Standort (Stadt)			
11	Meier	Hans- Hubert	München			
12	Huber	Rudolf	Stuttgart			
13	Müller	Beate	Hamburg			
14	Friedrichs	Karl Heinz	nicht bekannt			
15	von Klausewitz	Günther	Stuttgart			
16	Dr. Elbinger	Franz	Mannheim			
17						
18						
19						

Abbildung 3.25: Eine Zielmatrix mit der Funktion GLÄTTEN()

### 3.13 Die Funktion SÄUBERN()

Besonders Host-Daten haben die unangenehme Eigenschaft, allerlei Datenmüll mit sich herumschleppen. Berichte werden von Abrechnungssystemen auf Großrechnern bei der Herstellung von Textdateien oft mit Kopfinformationen, Druckersteuerbefehlen und Sonderzeichen versehen. Diese Zeichen machen nicht nur optisch auf dem Bildschirm, sondern oft auch auf dem Drucker Probleme. Je nach Druckermodell kann ein harmloses Zeichen in einer Zelle Seitenvorschübe, Tabulatorsprünge und Schriftwechsel verursachen.

Verwenden Sie die Funktion SÄUBERN(), um Tabellen von überflüssigen Zeichen zu befreien:

`=SÄUBERN(Text)`

Das Argument *Text* kann einen Zellbezug, einen Bereichsnamen oder einen Text in Anführungszeichen enthalten. Das Ergebnis der Funktion ist der Text ohne Sonderzeichen. Testen Sie die Funktion mit einem Text, der über die Funktion ZEICHEN() mit einem nicht druckbaren Sonderzeichen versehen ist:

A1: ="Umsatz " & ZEICHEN(7) & "1. Halbjahr 2005"  
 A2: =SÄUBERN(A1) Ergebnis: Umsatz 1. Halbjahr 2005




Abbildung 3.26: Die Funktion SÄUBERN() entfernt nicht druckbare Zeichen aus dem Text.

### 3.14 Die Funktion WIEDERHOLEN()

Mit dieser Funktion erzeugen Sie eine Textkette, die aus einer angegebenen Anzahl von Zeichen besteht. Die Funktionsklammern enthalten zwei Argumente:

=WIEDERHOLEN(Text;Multiplikator)

Das Argument *Text* steht für einen Zellbezug, einen Bereichsnamen oder einen Text in Anführungszeichen. Mit dem zweiten Argument *Multiplikator* geben Sie an, wie oft dieser Text in der Ergebniszelle wiederholt wird.

A1: @  
 A2: =WIEDERHOLEN(A1;5) Ergebnis: @@@@@

### 3.14.1 Praxisbeispiel: Artikelnummer mit Nullen auffüllen

Die vorliegende Liste enthält Artikel mit unterschiedlich großen Nummern. Ihre Aufgabe ist es, diese Nummern einheitlich mit 10 Stellen anzulegen. Kleinere Nummern füllen Sie dazu linksbündig mit Nullen auf.

Nr.	Produkt	Kategorie
23	Fujitsu Amilo	Notebook
240	HP XK Server	Personalcomputer
1200	DELL XL 500	Personalcomputer
32	DELL XP 3030	Personalcomputer
4500	Epson TX 80	Scanner
560090	Epson TX 80	Scanner
200	Gateway 1400	Personalcomputer
7900	HP DeskJet 560 C	Tintenstrahldrucker
3500	HP DeskJet 560 C	Tintenstrahldrucker
111	L 1000 CL	Laserdrucker
4500	L 1000 CL	Laserdrucker
450990	L 1000 CL	Laserdrucker
32009	Lexmark FX 300	Color-Laserdrucker

**Abbildung 3.27:** Die Nummern müssen auf eine Größe gebracht werden.

Mit einer Formatierung über das Zahlenformat ließe sich die Aufgabe relativ leicht lösen:

1. Markieren Sie die Spalte A, und wählen Sie FORMAT/ZELLEN/ZAHLEN/BENUTZERDEFINIERT.
2. Weisen Sie der Spalte das Zahlenformat zu, das alle Ziffern auf 10 Stellen bringt und dazu mit Nullen auffüllt:

0000000000





**Abbildung 3.28:** Das Zahlenformat füllt optisch mit Nullen.

Diese Lösung hat einen Haken: Wenn Sie die Tabelle in eine Access-Datenbank exportieren oder in eine Textdatei schreiben, verliert sie unter Umständen die Formatierung, und die Zahlen werden im Original ohne führende Nullen übernommen. Mit der WIEDERHOLEN()-Funktion erstellen Sie eine Nummer, die als Text in die Zelle übernommen wird und die Nullen sicher behält, auch beim Export in andere Formate:

1. Markieren Sie die Spalte B, und fügen Sie mit `[Strg] + [ + ]` eine neue Spalte ein.
2. Schreiben Sie in die erste Artikelzeile eine Formel, die die Länge der Zahl ermittelt:  
`B4: =LÄNGE(A4)`
3. Erstellen Sie die Funktion, die für den Abstand zwischen der Länge der Zahl und den vorgesehenen 10 Ziffern eine Reihe von Nullen produziert:  
`=WIEDERHOLEN("0";10-LÄNGE(A4))`
4. Jetzt können Sie die Zahl selbst noch an die Formel anfügen, verwenden Sie dazu das &-Zeichen:  
`=WIEDERHOLEN("0";10-LÄNGE(A4))&A4`

B4		=WIEDERHOLEN("0";10-LANGE(A4))&A4
A	B	
1	<b>Inventarliste</b>	
2		
3	<b>Nr.</b>	<b>Nr. neu</b>
4	23	000000023
5	240	0000000240
6	1200	0000001200
7	32	0000000032
8	4500	0000004500
9	560090	0000560090
10	200	0000000200
11	7900	0000007900
12	3500	0000003500
13	111	0000000111
14	4500	0000004500
15	450990	0000450990
16	32009	0000032009

Abbildung 3.29: Alle Nummern sind auf 10 Stellen aufgefüllt.

### 3.14.2 Praxisbeispiel: Balkenreihe per Funktion erzeugen

Zur Visualisierung von Größenverhältnissen, zeitlichen Strecken oder Differenzen verwenden Sie ein Balken- oder Säulendiagramm, erstellt mit dem Diagramm-Assistenten und als Diagrammobjekt oder Diagrammblatt in die Arbeitsmappe integriert. Eine kleine, weniger aufwendige Alternative bietet die WIEDERHOLEN()-Funktion: Erstellen Sie einfach eine Balkenreihe aus Sonderzeichen:

Der Projektplan zeigt Beginn und Ende der einzelnen Projektphasen an, berechnen Sie die Dauer in Wochen:

	A	R	C	D	F	F
1	<b>Projektplan</b>					
2		Beginn	Ende			
3	Phase 1: Konzept	01.01.2005	30.03.2005			
4	Phase 2: Design und Gestaltung	01.04.2005	21.04.2005			
5	Phase 3: Produktion	22.04.2005	31.08.2005			
6	Phase 4: Testlauf	01.09.2005	01.10.2005			
7	Phase 5: Installation	02.10.2005	01.11.2005			
8	Phase 6: Dokumentation	01.06.2005	01.11.2005			
9						
10						
11						
12						
13						
14						
15						
16						
17						
18						

Abbildung 3.30: Projektplan mit Phasen

D3: =(C3-B3)/7

1. Kopieren Sie die Formel auf die restlichen Zeilen, und weisen Sie der Spalte D über **FORMAT/ZELLEN/ZAHLEN/BENUTZERDEFINIERT** dieses Zahlenformat zu:

0" KW"

2. Erstellen Sie in der nächsten Spalte eine Formel, die den berechneten Wochenwert als Zeichenkette ausgibt:

=WIEDERHOLEN("n";D3)

3. Markieren Sie die Spalte D, und weisen Sie ihr über **FORMAT/ZELLEN/SCHRIFTART** die Schrift *WingDings* zu.

*WingDings* ist ein Zeichensatz, der die Buchstaben in Zeichen verwandelt, so auch das verwendete „n“, das jetzt als Kästchen für die Balkenreihe fungiert. Sie können die Schriftgröße der Spalte noch erhöhen, um den Effekt zu verstärken.




Abbildung 3.31: Mit WIEDERHOLEN und WingDings entsteht eine Balkenreihe.

## 3.15 Die Funktion FEST()

Mit dieser Funktion wandeln Sie eine Zahl in einen Text um und geben diesem dabei eine feste Anzahl Nachkommastellen mit.

**=FEST(Zahl;Dezimalstellen;Keine\_Punkte)**

Das Argument *Zahl* bezeichnet eine Zelle mit einer Zahl, einen Bereichsnamen, der auf eine Zahl verweist, oder eine Zahl. Mit *Dezimalstellen* bestimmen Sie die Anzahl der Dezimalstellen, die im Text rechts vom Dezimalkomma zu sehen sein sollen. Geben Sie eine negative Zahl ein, wird die Zahl links vom Komma auf diese Anzahl Stellen gerundet. Das Argument *Keine\_Punkte* bestimmt, ob im Text anschließend Tausenderpunkte gezeigt werden. Setzen Sie es auf WAHR, formatiert die Funktion den Text ohne Tausenderpunkte. Tragen Sie FALSCH ein oder lassen Sie das Argument unbesetzt, dann wird der Text Tausenderpunkte zeigen.

### Hinweis

Die in Text umgewandelte Zahl wird kaufmännisch gerundet (ab 5 an der letzten Stelle aufwärts, sonst abwärts).

A1: 1234,56789

A2: =FEST(A1;2) Ergebnis:

	A	B	C	D	E
1	Betrag	Betrag in Text			
2	1234,5678	1.234,57			
3	34567,8901	34.567,89			
4	120,4500	120,45			
5	99876,3401	99.876,34			
6	8024,5110	8.024,51			
7					
8					
9					
10					
11					
12					

**Abbildung 3.32:** Die Zahlen werden in Texte mit festen Nachkommastellen umgewandelt.

Die Funktion FEST() eignet sich besonders gut für Textverknüpfungen. Wenn Texte und Zahlenwerte gemeinsam auszugeben sind, fügen Sie diese mit &t zusammen und geben den String als Formel aus. Hier im Beispiel wird in der Tabelle eine gut sichtbare Meldung platziert, in der die Differenz der Kosten zum Budget ausgewiesen wird:

B8: =SUMME(A2:A6)

B9: 15.000

B10: ="Die Differenz zum veranschlagten Budget beträgt"&ZEICHEN(10)&FEST(B9-B8;2)&" Euro"

### Tipp

Mit der Funktion ZEICHEN(10) erzeugen Sie einen Zeilenumbruch im Text. Vergessen Sie nicht, die Zelle entsprechend zu formatieren (FORMAT/ZELLEN/AUSRICHTUNG).




Abbildung 3.33: Die Meldung zeigt eine Textkette, die Zahl wird mit FEST() eingebunden.

## 3.16 Die Funktion TEXT()

Ähnlich wie die Funktion FEST() wandelt auch diese Funktion Zahlen in Texte um, bietet aber mehr Komfort bei der Formatierung. Im Unterschied zu FEST() kann diese Funktion das Zahlenformat der Zahl direkt übernehmen.

=TEXT(Zahl;Textformat)

Das Argument *Zahl* bezeichnet eine Zelle mit einem Wert, einen Bereichsnamen oder eine Zahl. Im Argument *Textformat* geben Sie in Anführungszeichen das Format ein, das die in Text umgewandelte Zahl anschließend im Ergebnis anzeigen soll.

A1: 1234,567

A2: =TEXT(A1;"0,00") Ergebnis: 1234,57

### 3.16.1 Praxisbeispiel: Wochentage der Geburtstage ausgeben

In der Mitarbeiterliste der Personalabteilung stehen das Geburtsdatum des Mitarbeiters und das berechnete Alter. Der Wochentag, an dem er geboren ist oder an dem er dieses Jahr Geburtstag hat, lässt sich über das Zahlenformat einfach ermitteln.

	A	B	C
1	<b>Mitarbeiter</b>	<b>Geburtsdatum</b>	<b>Alter</b>
2	Hans Gruber	12.08.1978	26 Jahre
3	Michael Mitterhuber	31.12.1960	44 Jahre
4	Bern Braun	02.01.1981	24 Jahre
5	Gustav Gans	15.07.1968	36 Jahre

**Abbildung 3.34:** Die Mitarbeiterliste enthält das Geburtsdatum und das Alter.

Erstellen Sie ein Auswertungssystem, in dem der Geburtstag des Mitarbeiters in diesem Jahr ausführlich mit Wochentag in einer Meldung ausgegeben wird:

1. Markieren Sie die Mitarbeiterliste, und weisen Sie ihr mit EINFÜGEN/NAMEN/DEFINIEREN den Bereichsnamen MLISTE zu.
2. Setzen Sie den Zellzeiger in die Zelle A9, und wählen Sie DATEN/GÜLTIGKEIT.
3. Schalten Sie unter *Zulassen* um auf *Liste*, und tragen Sie als Quelle diese Formel ein, die alle Mitarbeiternamen beinhaltet:

=BEREICH.VERSCHIEBEN(MLISTE;1;;ZEILEN(MLISTE)-1;1)

4. Schreiben Sie das Datum des Geburtstags in diesem Jahr in die Tabelle:

B8: Geburtstag in diesem Jahr:

B9: =SVERWEIS(A9;MListe;2;FALSCH)

5. Erstellen Sie die Formel, die den Geburtstag noch einmal ausgibt, dieses Mal aber in einen Text eingebettet und mit der TEXT()-Funktion zur Wochentagsanzeige konvertiert:

B11: =A9&" hat am "&TEXT(B9;"TTTT, TT. MMMM")&" Geburtstag"




Abbildung 3.35: Das Auskunftssystem meldet auch den Wochentag des Geburtstags.

### 3.17 Die Funktionen ZEICHEN() und CODE()

Jedes Zeichen, das auf der Tastatur erzeugbar ist, hat einen bestimmten Zeichencode. Das Betriebssystem Windows ist ursprünglich auf Basis des ASCII-Codes entwickelt und später auf ANSI-Code umgestellt worden. Mit Windows XP kam Unicode, die derzeit aktuelle Zeichencodierung, die auch für das Internet gilt. In Excel arbeiten Sie mit dem ANSI-Zeichensatz.


<b>ASCII-Code (American Standard Code for Information Interchange)</b>	Älterer Zeichensatz, codierte maximal $2^7$ Zeichen = 128 Zeichen.
<b>ANSI-Code (American National Standard Institute)</b>	Der Zeichensatz für 8-Bit-Codierung ( $2^8 = 256$ Zeichen), mit dem die Office-Programme (Excel) arbeiten.
<b>Unicode</b>	Der Nachfolger des ANSI-Zeichensatzes, der alle fremdsprachigen Zeichen codiert. Unicode schließt die Schriftzeichen für die meisten Welt-sprachen mit ein. Mit $2^{32}$ Bit können exakt 4.294.967.296 Zeichen codiert werden.

**Tabelle 3.2:** Codierungssysteme für Zeichen

### 3.17.1 Zeichencodes feststellen

Excel bietet eine Übersicht über alle Zeichen und deren Codes, wahlweise in ASCII- oder Unicode-Codierung, an. Die Symbole bietet Excel ab der Version XP an:

1. Wählen Sie EINFÜGEN/SYMBOL.
2. Schalten Sie unter *Schriftart* auf die sSchrift um, mit der Sie in der Tabelle arbeiten.
3. Klicken Sie auf ein Zeichen, und überprüfen Sie in der Statusleiste der Dialogbox den Code.
4. Schalten Sie um auf die dezimale ASCII-Anzeige, um den Zeichencode dezimal zu sehen.



**Abbildung 3.36:** Die Symbolauswahl zeigt auch die Zeichencodes an.



Excel führt hier das ASCII-Format an, das hier gleichzusetzen ist mit ANSI (ANSI ist bis Code 127 identisch mit ASCII).

Die Standardanzeige der Codes ist Unicode (hex), die Zahlen werden in hexadezimaler Schreibweise angezeigt. Mit der Analyse-Funktion `HEX2DEC()` können Sie eine Hex-Zahl in eine Dezimalzahl umrechnen.

### 3.17.2 Codezahl eines Zeichens: `CODE()`

Die Funktion `CODE()` gibt die Codezahl eines Zeichens zurück und entspricht damit der Anzeige in der Statusleiste der Symbolbox:

`=CODE(Text)`

Das Argument *Text* steht für einen Zellbezug, einen Bereichsnamen oder einen Text in Anführungszeichen. Die Funktion gibt die Codezahl des ersten Buchstabens aus, auch wenn der Text mehr als eine Ziffer groß ist.

A1: Ø

A2: `=CODE(A1)` Ergebnis: 216

Geben Sie die Codezahlen der Ziffern von 0 bis 9 in einer Tabelle aus. Die Zeichenreihe erzeugen Sie mit der Zeilennummer über die Funktion `ZEILE()`.

A1: Zeichen

A2: `=ZEILE()-2`

B1: Code

B2: `=CODE(A2)`

Den Hex-Code berechnen Sie mit der Funktion `Hex2Dec()`. Damit dieser wie üblich vierstellig angezeigt wird, füllen Sie die Zelle über die Funktion `WIEDERHOLEN()` mit 0 auf.

C1: Hex-Code

C2: `=WIEDERHOLEN("0";4-LÄNGE(Dec2Hex(B2))&Dec2Hex(B2)`

	A	B	C
1	<b>Zeichen</b>	<b>Code</b>	<b>Hex-Code</b>
2	0	48	0030
3	1	49	0031
4	2	50	0032
5	3	51	0033
6	4	52	0034
7	5	53	0035
8	6	54	0036
9	7	55	0037
10	8	56	0038
11	9	57	0039

	A	B	C
1	<b>Zeichen</b>		
2	=ZEILE()-2		=DEC2HEX(B7)
3	=ZEILE()-2		=DEC2HEX(B8)
4	=ZEILE()-2		=DEC2HEX(B9)
5	=ZEILE()-2		=DEC2HEX(B5)
6	=ZEILE()-2		=DEC2HEX(B6)
7	=ZEILE()-2		=DEC2HEX(B1)
8	=ZEILE()-2		=DEC2HEX(B9)
9	=ZEILE()-2		=DEC2HEX(B9)
10	=ZEILE()-2	=CODE(A10)	=WIEDERHOLEN("0";4-LÄNGE(DEC2HEX(B10)))&DEC2HEX(B10)
11	=ZEILE()-2	=CODE(A11)	=WIEDERHOLEN("0";4-LÄNGE(DEC2HEX(B11)))&DEC2HEX(B11)

**Abbildung 3.37:** Die Funktion CODE() berechnet die Codezahl eines Zeichens.

### 3.17.3 Praxisbeispiel: Numerische und alphanumerische Einträge sortieren

Excel sortiert Zellinhalte alphanumerisch und orientiert sich dabei an der Codierung der Zeichen. Mit der Funktion CODE() können Sie diese Codierung ermitteln:

Zeichenbereich	Codierung
0–9	48–57
A–Z	65–90
a–z	97–122

**Tabelle 3.3:** Codierungsgruppen im ANSI-Code

Die übrigen Zeichen sind Sonderzeichen aller Art, im ANSI-Code bis zur Codezahl 255 codiert. Wenn Sie eine Tabelle sortieren, gilt die Codezahl als Kriterium, und da numerische Codes (Zahlen) kleiner sind als alphanumerische (Buchstaben), werden zuerst Zahlen, dann Buchstaben eingeordnet.

In der Praxis werden Sie auf Sonderfälle stoßen, die sich nicht nach diesem Schema sortieren lassen. Hier ein Beispiel: Die Inhaltsliste eines Manuskripts ist mit einer speziellen Nummerierung versehen. Wenn Sie die Liste nach der ersten Spalte sortieren, werden alle numerischen Einträge nach oben geschoben, dann folgen die Zahlen/Buchstaben-Kombinationen:

	A	B
1	1	Vorwort
2	1a	Über den Autor
3	1b	Inhaltsverzeichnis
4	1c	Impressum
5	2	Manuskript
6	3	Anhang
7	3a	Linkliste
8	3b	Stichwortverzeichnis
9	3c	Fachwörterbuch

	A	B
1	1	Vorwort
2	2	Manuskript
3	3	Anhang
4	1a	Über den Autor
5	1b	Inhaltsverzeichnis
6	1c	Impressum
7	3a	Linkliste
8	3b	Stichwortverzeichnis
9	3c	Fachwörterbuch

**Abbildung 3.38:** Numerische und alphanumerische Einträge werden falsch sortiert.

Schreiben Sie in der nächsten freien Spalte diese Formel:

```
C1: =WENN(CODE(RECHTS(A1;1)) > 64;
    LINKS(A1;LÄNGE(A1)-1)*1000+CODE(RECHTS(KLEIN(A1);1))+
    WENN(CODE(RECHTS(A1;1)) < 97;0,5;0);A1*1000)
```

Die Formel prüft das Codezeichen der letzten Ziffer und erstellt eine Sortierzahl mit der Codenummer. Sortieren Sie die Liste nach dieser Spalte, werden die Einträge wieder richtig eingeordnet.

	A	B	C1
1	1	Vorwort	1000
2	1a	Über den Autor	1097
3	1b	Inhaltsverzeichnis	1098
4	1c	Impressum	1099
5	2	Manuskript	2000
6	3	Anhang	3000
7	3a	Linkliste	3097
8	3b	Stichwortverzeichnis	3098
9	3c	Fachwörterbuch	3099
10			
11			
12			
13			
14			
15			

**Abbildung 3.39:** Mit der Codeprüfung wird die Liste korrekt sortiert.

### 3.17.4 Das Zeichen einer Codezahl: ZEICHEN()

Der umgekehrte Weg: Sie haben eine (dezimale) Codezahl, ermitteln Sie das dazugehörige Zeichen. Die Funktion ZEICHEN() zeigt Ihnen, welches Zeichen zu einer Codezahl gehört:

=ZEICHEN(Zahl)

Das Argument *Zahl* steht für einen Zellbezug, einen Bereichnamen, der auf eine Zahl verweist, oder eine Zahl selbst. Das Ergebnis ist das Zeichen in Textform.

Erstellen Sie eine ANSI-Codetabelle mit allen druckbaren Zeichen im Codebereich 32–255. Schreiben Sie die Zahlen in die Tabelle oder berechnen Sie sie mit der Funktion ZEILE() aus der Zeilennummer. Tragen Sie die Funktion ZEICHEN() jeweils neben der Codezahl ein.

The screenshot shows an Excel spreadsheet titled "Microsoft Excel - FF Textfunktionen Beispiele.xls". The formula bar displays "=ZEICHEN(D5)". The spreadsheet contains an ANSI-Codetabelle with columns for 'Wert' and 'Code' for each character. The character '!' is highlighted in cell D5.

	Wert	Code	Wert	Code	Wert	Code	Wert	Code	Wert	Code	Wert	Code	Wert	Code	Wert	Code
1	ANSI-Codes															
2																
3	Wert	Code	Wert	Code	Wert	Code	Wert	Code	Wert	Code	Wert	Code	Wert	Code	Wert	Code
4	32		63	?	94	^	125	}	156	œ	187	»	218	Ù	249	
5	33	!	64	⊗	95	~	126	~	157	ð	188	¼	219	Ú	250	
6	34	"	65	A	96	`	127	␣	158	z	189	½	220	Û	251	
7	35	#	66	B	97	a	128	€	159	ÿ	190	¾	221	Ü	252	
8	36	\$	67	C	98	b	129	␣	160		191	¸	222	ß	253	
9	37	%	68	D	99	c	130	,	161	j	192	À	223	ä	254	
10	38	&	69	E	100	d	131	f	162	ç	193	Á	224	å	255	
11	39	'	70	F	101	e	132	"	163	£	194	Â	225	ä		
12	40	(	71	G	102	f	133	...	164	=	195	Ã	226	å		
13	41	)	72	H	103	g	134	†	165	¥	196	Ä	227	ä		
14	42	*	73	I	104	h	135	‡	166	!	197	Å	228	a		
15	43	+	74	J	105	i	136	^	167	\$	198	Æ	229	ä		
16	44	,	75	K	106	j	137	‰	168		199	Ç	230	æ		
17	45	-	76	L	107	k	138	Š	169	©	200	È	231	ç		
18	46	.	77	M	108	l	139	<	170	*	201	É	232	è		
19	47	/	78	N	109	m	140	œ	171	°	202	Ê	233	é		
20	48	U	79	O	110	n	141	␣	172	¬	203	Ë	234	e		
21	49	1	80	P	111	o	142	Ž	173	-	204	Ì	235	e		
22	50	2	81	Q	112	p	143	␣	174	®	205	Í	236	i		

Abbildung 3.40: Eine ANSI-Codetabelle mit der Funktion ZEICHEN()

### 3.18 Die Funktion DM()

Der Name dieser Funktion ist abhängig von den Ländereinstellungen, und auch wenn es bei uns seit 2002 keine Deutsche Mark mehr gibt, steht sie immer noch zur Verfügung. DM() liefert einen Text mit Währungsformat.

**=DM(Zahl)**

*Zahl* ist die umzuwandelnde Zahl. Die Rundung erfolgt auf die durch das Argument *Dezimalstellen* angegebene Stellenanzahl. Werden keine Dezimalstellen angegeben, erfolgt die Ausgabe standardmäßig mit zwei Stellen hinter dem Komma. Dem Text wird rechts ein Leerzeichen und das Währungssymbol angehängt, das in der Systemsteuerung für Währungen eingestellt ist (Standard: EUR oder €).

DM(56,3457)	=	56,34 €
DM(-2;2)	=	2,00 €
DM(0;5)	=	0,00000 €